NATIONAL BUREAU OF STANDARDS REPORT

10 494

FIRE ENDURANCE THERMAL ANALYSIS OF CONSTRUCTION WALLS

NBS

U.S. DEPARTMENT OF COMMERCE

NATIONAL BUREAU OF STANDARDS

NATIONAL BUREAU OF STANDARDS

The National Bureau of Standards [1] was established by an act of Congress March 3, 1901. Today, in addition to serving as the Nation's central measurement laboratory, the Bureau is a principal focal point in the Federal Government for assuring maximum application of the physical and engineering sciences to the advancement of technology in industry and commerce. To this end the Bureau conducts research and provides central national services in four broad program areas. These are: (1) basic measurements and standards, (2) materials measurements and standards, (3) technological measurements and standards, and (4) transfer of technology.

The Bureau comprises the Institute for Basic Standards, the Institute for Materials Research, the Institute for Applied Technology, the Center for Radiation Research, the Center for Computer Sciences and Technology, and the Office for Information Programs.

THE INSTITUTE FOR BASIC STANDARDS provides the central basis within the United States of a complete and consistent system of physical measurement; coordinates that system with measurement systems of other nations; and furnishes essential services leading to accurate and uniform physical measurements throughout the Nation's scientific community, industry, and commerce. The Institute consists of an Office of Measurement Services and the following technical divisions:

Applied Mathematics—Electricity—Metrology—Mechanics—Heat—Atomic and Molecular Physics—Radio Physics [2]—Radio Engineering [2]—Time and Frequency [2]—Astrophysics [2]—Cryogenics.[2]

THE INSTITUTE FOR MATERIALS RESEARCH conducts materials research leading to improved methods of measurement standards, and data on the properties of well-characterized materials needed by industry, commerce, educational institutions, and Government; develops, produces, and distributes standard reference materials; relates the physical and chemical properties of materials to their behavior and their interaction with their environments; and provides advisory and research services to other Government agencies. The Institute consists of an Office of Standard Reference Materials and the following divisions:

Analytical Chemistry—Polymers—Metallurgy—Inorganic Materials—Physical Chemistry.

THE INSTITUTE FOR APPLIED TECHNOLOGY provides technical services to promote the use of available technology and to facilitate technological innovation in industry and Government; cooperates with public and private organizations in the development of technological standards, and test methodologies; and provides advisory and research services for Federal, state, and local government agencies. The Institute consists of the following technical divisions and offices:

Engineering Standards—Weights and Measures — Invention and Innovation — Vehicle Systems Research—Product Evaluation—Building Research—Instrument Shops—Measurement Engineering—Electronic Technology—Technical Analysis.

THE CENTER FOR RADIATION RESEARCH engages in research, measurement, and application of radiation to the solution of Bureau mission problems and the problems of other agencies and institutions. The Center consists of the following divisions:

Reactor Radiation—Linac Radiation—Nuclear Radiation—Applied Radiation.

THE CENTER FOR COMPUTER SCIENCES AND TECHNOLOGY conducts research and provides technical services designed to aid Government agencies in the selection, acquisition, and effective use of automatic data processing equipment; and serves as the principal focus for the development of Federal standards for automatic data processing equipment, techniques, and computer languages. The Center consists of the following offices and divisions:

Information Processing Standards—Computer Information — Computer Services — Systems Development—Information Processing Technology.

THE OFFICE FOR INFORMATION PROGRAMS promotes optimum dissemination and accessibility of scientific information generated within NBS and other agencies of the Federal government; promotes the development of the National Standard Reference Data System and a system of information analysis centers dealing with the broader aspects of the National Measurement System, and provides appropriate services to ensure that the NBS staff has optimum accessibility to the scientific information of the world. The Office consists of the following organizational units:

Office of Standard Reference Data—Clearinghouse for Federal Scientific and Technical Information [3]—Office of Technical Information and Publications—Library—Office of Public Information—Office of International Relations.

[1] Headquarters and Laboratories at Gaithersburg, Maryland, unless otherwise noted; mailing address Washington, D.C. 20234.
[2] Located at Boulder, Colorado 80302.
[3] Located at 5285 Port Royal Road, Springfield, Virginia 22151.

NATIONAL BUREAU OF STANDARDS REPORT

NBS PROJECT
4219112

September 29, 1971

NBS REPORT
10 494

FIRE ENDURANCE THERMAL ANALYSIS OF CONSTRUCTION WALLS

by

Francis C. W. Fung
Fire Research Section
Building Research Division
Institute for Applied Technology
National Bureau of Standards
Washington, D. C. 20234

IMPORTANT NOTICE

U.S. DEPARTMENT OF COMMERCE

NATIONAL BUREAU OF STANDARDS

Table of Contents

ABSTRACT

A general one dimensional transient heat transfer numerical program has been developed for composite building constructions with arbitrary air gap locations. The complete Fortran language program as used on the NBS Univac 1108 computer is given. A discussion of the program and instruction for its use are facilitated by the aid of examples. Numerical solutions using the present program compare favorably with experimental data in standard fire endurance tests.

1.0 INTRODUCTION

The fire performance of building constructions is generally evaluated by a large scale laboratory fire endurance test (ASTM E 119) in which one surface is exposed to fire, controlled according to a prescribed increasing temperature history simulating the burnout of combustibles. The fire endurance rating of the construction is the time period during which it withstands the fire exposure without (a) structural failure, (b) the development of cracks through which flames can cross, or (c) the temperature rise on the unexposed surface exceeding a prescribed limit (250^oF rise average, 325^oF rise at a single point). Where the failure criterion is due to heat transmission without complications due to structural or physical effects, heat transfer analysis should provide a means for prediction and design.

A particular aspect of the fire endurance test which is not well defined but which probably plays a significant role in fire performance, is the effect of mass flow (air and combustion gases) due to pressure differences since typical building constructions consist of a series of composite layers and intermediate air layers, a transient heat and air infiltration model was formulated. The program is particularly suitable for evaluating the thermal fire endurance of building constructions where various combinations of solid-to-solid and solid-to-air contacts are encountered. For each solid layer, the present program has provisions for phase changes, heat generation and absorption, and thermal property variations commonly found in building materials. Through the air spaces the modes

of heat transfer include radiation and convection with temperature-dependent heat transfer coefficients and air properties.

A number of analog and numerical programs for fire endurance evaluations had been in existence for some time [1, 2, 3, 4]. A more flexible and general finite difference program was developed by Krokosky as recently as 1970 [5]. For a review of the existing thermal analysers for fire endurance evaluation one is thus referred to [4 and 5].

The present numerical program was developed to incorporate into fire endurance analyses the following features which are desirable and yet not readily available in existing programs:

1. Options to handle heat exposure on one or both sides of structure.
2. Heat balances in air spaces to allow for air infiltration, and heat generation and absorption in air spaces.
3. For ease of application to building structures an input card, say 101101, is sufficient to instruct the computer of the specified number of solid layers and air spaces in a given problem.
4. Temperature-dependent properties and known chemical heat exchanges of various building materials are stored in a subroutine and called by an input card, say 2331, where the numbers indicate coded materials stored in the subroutine.
5. Duration as well as amount of known chemical heat exchanges can be varied in any material.

2.0 GOVERNING EQUATION AND BOUNDARY CONDITIONS

The governing equation for one dimensional transient heat flow is the well known heat diffusion equation. Including a term for internal heat

generation, this can be expressed as,

$$\frac{\partial}{\partial x}\left(k\frac{\partial T}{\partial k}\right) + g = \rho c\frac{\partial T}{\partial \theta} \qquad (1)$$

where,

T= absolute temperature in solid.

x = coordinate in direction of heat flow.

k = heat conduction coefficient.

g = time rate of heat generation per unit volume in solid.

ρ= density of solid.

c = specific heat capacity of solid.

θ= time

Boundary Conditions [6, 7] :

The following types of boundary conditions need to be evaluated for the general problem;

(a) solid to air

(b) energy balance in air space

(c) air to solid

(d) solid to surrounding

(e) solid to symmetry plane

(f) solid to solid

(g) furnace gases to first surface layer.

Figure 1



(a) Solid to Air (See Figure 1)

$$\frac{\rho_1 c_1 \Delta x_1}{2} \frac{\partial T_1}{\partial \theta} = -k_1 \frac{\partial T_1}{\partial x} - \sigma \epsilon_{12} (T_1^4 - TA^4) - h_1 (T_1 - TA) + g_1 \frac{\Delta x_1}{2} \quad (2)$$

(b) Energy Balance in Air Space (See Figure 1)

$$\ell_a \, c_{pa} \, \rho_a \frac{\partial TA}{\partial \theta} = h_1 (T_1 - TA) - h_2 (TA - T_2) + \mathring{m} \, c_{pa} (T_1 - T_2) + \ell_a \, g_a \quad (3)$$

(c) Air to Solid (See Figure 1)

$$\frac{\rho_2 c_2 \Delta x_2}{2} \frac{\partial T_2}{\partial \theta} = k_2 \frac{\partial T_2}{\partial x} + \sigma \epsilon_{12} (TA^4 - T_2^4) + h_2 (TA - T_2) + g_2 \frac{\Delta x_2}{2} \quad (4)$$

In the above equations the subscripts 1 and 2 indicate solid 1 and solid 2 respectively as shown in Figure 1, and where $\epsilon_{12} = \dfrac{1}{\dfrac{1}{\epsilon_1} + \dfrac{1}{\epsilon_2} - 1}$

and ϵ_1 , ϵ_2 are emissivity of surface 1 and surface 2,

T_2, T_1: surface temperatures as shown in Figure 1

Δx_1, Δx_2: incremental length in x direction

σ: Stefan Boltzman constant

TA: temperature of air

$h_1 = K\left(\dfrac{T_1 - TA}{L}\right)^{\frac{1}{4}}$: convection heat transfer coefficient from solid to air

L: characteristic dimension of panel

k_1, k_2: heat conduction coefficient in solid 1, and 2

g_1, g_2: time rate of heat generation or absorption per unit volume in solid 1 and 2

ℓ_a: air gap spacing

c_{pa}: specific heat capacity of air

ρ_a: density of air

$h_2 = K\left(\dfrac{TA - T_2}{L}\right)^{\frac{1}{4}}$: convection heat transfer coefficient from air to solid

$\dot{m}$: rate of mass transfer per unit area due to pressure difference

g_a: rate of heat generation per unit volume in air space due to combustibles.

K: an empirical convection heat transfer constant, K=.27 for vertical surfaces, and K=.38 for horizontal surfaces.

Figure 2 -



(d) <u>Solid to Ambient (See Figure 2)</u>

$$\frac{\rho c\Delta x}{2}\frac{\partial T}{\partial \theta} = -k\frac{\partial T}{\partial \theta} - \sigma\epsilon(T^4 - T_o^4) - h_o(T - T_o) + g\frac{\Delta x}{2} \quad (6)$$

where the additional variables are,

T_o: ambient temperature

$h_o = K\left(\frac{T - T_o}{L}\right)^{\frac{1}{4}}$: convection heat transfer coefficient

ϵ: emissivity of surface

T: surface temperature of solid

Figure 3 -



(e) <u>Solid to Symmetry Plane (See Figure 3)</u>

By symmetry, temperature on both solid surfaces that face each other are equal, however $T \geq T_s$. So there will be heat transfer from solid to air. At the interface we have,

$$\frac{\rho c\Delta x}{2}\frac{\partial T}{\partial \theta} = -k\frac{\partial T}{\partial x} - h_s(T - T_s) + g\frac{\Delta x}{2} - \sigma\epsilon(T^4 - T_s^4) \quad (8a)$$

At the air space we have,

$$\ell_s c_{pa} \rho_a \frac{\partial T_s}{\partial \theta} = h_s (T - T_s) + \sigma\varepsilon (T^4 - T_s^4) \qquad (8b)$$

where,

T: surface temperature of solids

T_s: temperature of air in symmetry plane

ℓ_s: distance from solid to symmetry plane

$h_s = K\left(\frac{T - T_s}{L}\right)^{\frac{1}{4}}$: convection heat transfer coefficient from solid to symmetry plane.

Figure 4 -



(f) <u>Solid to Solid (See Figure 4)</u>

Consider interface between Solid 1 and Solid 2 as shown in Figure 4. Let T_1 and T_2 be the interface temperature in each solid respectively. A heat balance for the interface node can be expressed as, $T_2 = T_1$,

$$\rho_1 c_1 \frac{\Delta x_1}{2} \frac{\partial T_1}{\partial \theta} + \rho_2 c_2 \frac{\Delta x_2}{2} \frac{\partial T_2}{\partial \theta} = g_1 \frac{\Delta x_1}{2} + g_2 \frac{\Delta x_2}{2} - k_1 \frac{\partial T_1}{\partial x} + k_2 \frac{\partial T_2}{\partial x} \qquad (5)$$

where subscript indicates conditions in solid layer 1 or 2.

Figure 5 -



(g) <u>Furnace Gases to First Solid Surface (See Figure 5)</u>

Consider heat transfer from furnace gases to first solid surface. The main modes of heat transfer will be radiation and convection.

A good approximation of ASTM E 119 fire curve is given by the following three formulas, (where T = temperature in degrees C, and Θ = time in minutes);

$$T = 940 \frac{\Theta}{\Theta + 4} \,^{o}C + 20^{o}C \qquad 0 < \Theta < 50 \text{ min (error} \pm 2\%)$$

$$T = 926^{o}C + 0.7\Theta^{o}C - 0.0131(120 - \Theta)^2 \,^{o}C \qquad 50 \text{ min} < \Theta < 115 \text{ min (no appreciable error)}$$

$$T = 926^{o}C + 0.7\Theta^{o}C \qquad 115 \text{ min} < \Theta < 480 \text{ min (linear exact)}$$

Heat balance from furnace to first solid surface,

$$\frac{\rho c \Delta x}{2} \frac{\partial T_1}{\partial \Theta} = k \frac{\partial T}{\partial x} + \sigma\varepsilon(T_F^4 - T_1^4) + h_F(T_F - T_1) + g \frac{\Delta x}{2} \qquad (9)$$

where T_F is absolute furnace temperature, and

$$h_F = K\left(\frac{T_F - T_1}{L}\right)^{\frac{1}{4}}$$

3.0 PROBLEM FORMULATION AND FINITE DIFFERENCE EQUATIONS

Consider general one-dimensional heat transfer problem containing N solid layers and m air spaces in any order. To facilitate formulation and discussion let's introduce i as the running index for air spaces. In Figure 6, a general multi-layer configuration is shown, where numbers in circles indicate the applicable equation at the given node as discussed in the previous section.



Figure 6. A Composite Wall with N Layers

Note: Number in circle indicates applicable equation as discussed in Section 2.0.

To solve our problem numerically with the nonlinear heat diffusion equation and associated complex system of boundary conditions we shall use forward time differencing and central space differencing scheme. Furthermore, we must use three subscripts.

Let:

$T_{i,j}$: Temperature of jth node in ith solid

i: 1...N (number of solid layers)

j: 1...n (number of nodes on solid layer)

k: 1...m (number of air spaces)

See Figure 7 for finite differencing network.

Figure 7. Sketch of Finite Differencing Network in a Solid Layer



Furthermore, let

TA_k: Temperature of kth air spacing

Θ: Time

$\Delta\Theta$: Time increment

$T'_{i,j}$: Temperature at $(\Theta + \Delta\Theta)$ of jth node in ith solid

ℓ_i: Thickness of ith solid

ℓA_k: Thickness of kth air spacing

$\Delta x_i = \frac{\ell_i}{n - 1}$: Difference spacing for ith solid

$g_{i,j}$: rate of heat generation per unit volume

$k_{i,j}$: heat conduction coefficient

α_i: heat diffusion coefficient

ga_k: rate of heat generation per unit volume in air space

$\mathring{m}_k$: rate of mass transfer per unit area

Applying our finite differencing scheme we have the following general finite difference expressions corresponding to the previously discussed governing equation and boundary conditions; [6, 7, and 8]

<u>Governing equation</u>

$$T'_{ij} = \frac{1}{M_i}(T_{i(j - i)} + T_{i(j + i)}) + (1 - \frac{2}{M_i})T_{ij} + \frac{g_{ij}\Delta x_i^2}{k_{ij}M_i} \qquad (1)$$

where $M_i = \frac{(\Delta x_i)^2}{\alpha_i \Delta\Theta}$

From equation (1) we require $M_i > 2$ for stability.

<u>Solid to Air</u>,

$$T'_{i,n} = T_{i,n} + \frac{2}{M_i}(T_{i,n-1} - T_{i,n}) - R_i[(T_{i,n})^4 - (T_{i+1,1})^4] - H_i(T_{i,n} - TA_k)^{\frac{5}{4}} + \frac{g_{i,n}\Delta x_i^2}{k_{i,n}M_i} \qquad (2)$$

where

$$R_i = \frac{2\Delta\theta\sigma\epsilon_{12}}{\Delta x_i \rho_i c_i}$$

c_i: specific heat of ith layer.

$$\epsilon_{12} = \frac{1}{\frac{1}{\epsilon_1} + \frac{1}{\epsilon_2} - 1}$$: where ϵ_1 and ϵ_2 are emmisivity of two parallel solid layers.

$$H_i = \frac{2\Delta\theta K}{\rho_i c_i \Delta x_i (L)^{\frac{1}{4}}}$$

L: characteristic dimension of panel.

Air Space Heat Balance

$$TA_k' = TA_k + P_k(T_{i,n} - TA_k)^{\frac{5}{4}} - Q_K(TA_k - T_{i+1,1})^{\frac{5}{4}} + S_k(T_{i,n} - T_{i+1,1}) + GA_k \tag{3}$$

where

$$P_k = \left(\frac{K}{L^{\frac{1}{4}}}\right)\frac{\Delta\theta}{\ell A_k c_{pa}\rho_a}$$

ρ_a: air density function of temperature.

c_{pa}: specific heat of air.

$$Q_K = \left(\frac{K}{L^{\frac{1}{4}}}\right)\frac{\Delta\theta}{\ell A_k c_{pa}\rho_a}$$

$$S_k = \frac{\dot{m}\Delta\theta}{\ell A_k \rho_a}$$

$$GA_k = \frac{ga_k\Delta\theta}{c_{pa}\rho_a}$$

Air to Solid

$$T'_{i+1,1} = T_{i+1,1} + R_i[(T_{in})^4 - (T_{i+1,1})^4] - \frac{2}{M_{i+1}}(T_{i+1,1} - T_{i+1,2}) + H_{i+1}(TA_k - T_{i+1,1})^{\frac{5}{4}} \tag{4}$$

R_i, M_{i+1}, H_{i+1} as defined before.

Solid to Ambient

$$T'_{N,n} = T_{N,n} + \frac{2}{M_N}(T_{N,n-1} - T_{N,n}) - R_N(T_{N,n}^4 - T_o^4) - H_N(T_{N,n} - T_o)^{\frac{5}{4}} + \frac{G_{N,n}(\Delta x_N)^2}{k_{N,n}M_N} \tag{6}$$

where

$$R_N = \frac{2\Delta\theta\sigma\epsilon_N}{\Delta x_N \rho_N c_N}$$

ϵ_N: emisivity of solid (last node to ambient).

H_N, $k_{N,n}$ and M_N as defined before

T_o: ambient temperature

Solid to Symmetry Plane

$$T'_{N,n} = T_{N,n} + \frac{2}{M_N}(T_{N,n-1} - T_{N,n}) - H_N(T_{N,n} - T_s)^{\frac{5}{4}} + \frac{G_{N,n}(\Delta x_N)^2}{k_{N,n}M_N} - R_N(T_{N,n}^4 - T_s^4) \tag{8a}$$

$$T'_s = T_s + P_s(T_{N,n} - T_s)^{\frac{5}{4}} + R_N(T_{N,n}^4 - T_s^4) \tag{8b}$$

where

T_s: temperature at symmetry plane

$$P_s = \left(\frac{K}{L^{\frac{1}{4}}}\right) \frac{\Delta\theta}{\ell_s c_{pa} \rho_a}$$

ℓ_s: distance between solid surface and symmetry plane.

Solid to Solid

$$T'_{i+1,1} = T'_{i,n}$$

$$T'_{i,n} = [A_i T_{i,n} + A_{i+1} T_{i+1,1} + B_i g_{i,n} + B_{i+1} g_{i+1,1} + D_{i,n}(T_{i,n-1} - T_{i,n}) - D_{i+1,1}(T_{i+1,1} - T_{i+1,2})](A_i + A_{i+1}) \quad (5)$$

where

$$A_i = \frac{\rho_i c_i \Delta x_i}{2}$$

$$B_i = \frac{\Delta x_i \Delta\theta}{2}$$

$$D_{i,n} = \frac{k_{i,n} \Delta\theta}{\Delta x_i}$$

$$D_{i+1,1} = \frac{k_{i+1,1} \Delta\theta}{\Delta x_{i+1}}$$

$g_{i,j}$: rate of heat generation per unit volume in ith solid.

<u>Furnace Gases to First Solid Surface</u>

$$T'_{1,1} = T_{1,1} + R_1 (T_F^4 - T_{1,1}^4) + H_1 (T_F - T_{1,1}) + \frac{G_{1,1} \Delta x_1}{k_{1,1} M_1} - \frac{2}{M_1}(T_{1,1} - T_{1,2}) \quad (9)$$

where

$$R_1 = \frac{2\Delta\theta\sigma\varepsilon_o}{\Delta x_1 \rho_1 c_1}$$

ε_o: emisivity of first surface (furnace gases to 1st solid).

T_F: furnace temperature

$$H_1 = \frac{K}{L^{\frac{1}{4}}} \; \frac{2\Delta\theta}{\rho_1 c_1 \Delta x_1}$$

4.0 DISCUSSION OF NUMERICAL PROGRAM

A flow chart of the main program is presented in Figure 7. A listing of the complete numerical program is also presented at the end of this report. The logical sequence of the main program can be grasped readily by first considering the flow chart.

A list of input parameters is read in on data cards (examples of which are shown at Section 5.3) as follows:

NN= number of solid layers

N= number of nodes in each layer

M= number of air spaces

ID= sequence of numbers 1 or 0, specifying the sequence of solid layers and air spaces, e.g., 101011101 means solid-air-solid-air-solid-solid-solid-air-solid.

IDD= sequence of numbers specifying the material of each solid layer, e.g., 122321 means the six solid layers of the problem are of materials 1, 2, 2, 3, 2, 1 in that order. Various temperature dependent material properties are stored in Subroutine Prop.

AL(I): thickness of ith solid layer in feet

ALPHA(I): heat diffusion coefficient of ith solid layer, ft^2/hr.

(ALPHA assumed constant other temperature dependent thermal properties stored in Subroutine Prop.

RHO(I): density of ith solid layer in lb/ft^3

AI(I): thickness of air gap spacing in feet

GA(k): rate of heat generation per unit volume due to combustibles in kth air space, Btu/ft^3-hr.

DTHETA: time increment in fraction of hour.

AMO: Mass flux through walls in lb/ft^2-hr

GM1, GM2, GM3 etc: Heat absorption or release due to phase change in material 1, 2, or 3 etc., Btu/ft^3hr.

Subroutine Prop: stores temperature dependent thermal properties and phase change reactions of some common building materials. This subroutine can be expanded as desire when new materials are encountered. When called from the main program this subroutine supplies the thermal properties for the ith layer of solid currently being calculated. The function subroutines are self explanatory. The comment statement at the beginning of each function subroutine properly identifies it with the corresponding equation and boundary conditions in section 3.0.

5.0 SAMPLE CASES CALCULATED

5.1 Description of Panels

Case 1: Four solid layers (gypsum, wood, wood, gypsum), three air gaps, heat from one side as shown:



Case 2: Four solid layers (gypsum, wood, wood, gypsum), three air gaps, heating from both sides as shown:



Case 3: One solid layer (plywood either 1/2" or 5/8" thick), heating from one side of panel.

Case 4: Two solid layers and one air gap (Brick - Air - Brick), heating from one side as shown:



5.2 Thermal Properties

The following thermal properties taken from Ref. 1 and Ref. 2 are used in all calculations, $\epsilon = .9$ for all surfaces.

Gypsum Board

$\alpha = .008\ ft^2/hr$

$\rho = 60\ lb/ft^3$

$c = .26$ Btu/lb°F

$K = .125$ Btu/hr ft°F $0 < T < 200°F$

$K = .075$ Btu/hr ft°F $200°F < T < 400°F$

$K = (.05+\frac{T}{16,000})$ Btu/hr ft°F   400°F<T<2000°F

Heat of moisture desorption and calcination: 20,740 Btu/ft^3

Plywood

α = .006 ft^2/hr

ρ = 31 lb/ft^3

c = .67 Btu/lb°F

K = .065 Btu/hr ft°F

Heat of moisture desorption: 15 00 Btu/ft^3

Brick

α = .028 ft^2/hr

ρ = 110 lb/ft^3

c = .216 Btu/lb°F

K =1.0 Btu/hr ft°F   0<T<200°F

K = .46 + 2T/10,000 Btu/hr ft°F   200°F<T<2000°F

Heat of moisture desorption = 5800 Btu/ft^3

5.3 Input Data

The following are print outs of input data for the sample cases calculated:

Case 1: Four solid layers, 3 air gaps, furnace on one side of panel. (gypsum - air - wood - air - wood - air - gypsum)

```
   4    5    3    1     68.000        .000     5380.000    1500.000
1  0  1  0  1  0  1
1  3  3  1
            .250         .000         .250         .000
            .250         .000
            .052         .008       60.000         .260
            .042         .006       31.000         .670
            .042         .006       31.000         .670
            .052         .008       60.000         .260
```

Case 2: Four solid layers, 3 air gaps furnace on both sides of panel.

(Note: Symmetry has been invoked gypsum - air - wood - symmetry plane)

```
   2     5     1     2     68.000          .000   5380.000   2250.000
1  0  1
1  3
          .500          .000
          .052          .008        60.000          .260
          .042          .006        31.000          .670
```

Case 3: One solid layer (plywood) furnace on one side of panel.

```
   1     5     0     1     68.000    .00    5800.000    1500.000
1
3
        .042        .006          31.000        .670
```

Case 4: Two solid layers and one air gap (brick - air - brick) furnace on one side of wall.

```
   2     25     1     1     68.000     .00     5800.000     1500.000
1  0  1
2  2
          .5            .000
          .313          .023        110.000        .216
          .313          .023        110.000        .216
```

Comparison plots of calculation and standard fire endurance tests are presented in fig. 8, 9, and 10. Tests were conducted at NBS, Washington, D.C. and National Gypsum Corporation, Buffalo, N.Y.

Finally a word of caution in the use of a thermal analyzer program with air gap heat balances such as the present program. Due to the low heat capacity of air, one must exercise caution in selecting the incremental $\Delta\theta$ and ΔX such that the air gap temperature will not rise above the temperature of the node preceeding it. From experience when this occurs

instability will set in. The solution is then to decrease $\Delta\theta$ or ΔX and at the same time maintaining the criteria $M>2$. The time increment $\Delta\theta$ is 5 seconds and the number of nodes is 5 in all calculated cases except case 4 where $\Delta\theta$ is decreased to 2.5 seconds and n is increased to 25 to avoid the instability problem discussed in above.



Fig. 8 Temperature History of Case 1 (4 Layers, 3 Air Gaps, Gypsum-Wood-Wood-Gypsum, Heating from one side).



Fig. 9 Temperature History of Case 2 (4 Layers, 3 Air Gaps, Gypsum-Wood-Wood-Gypsum, Heating from both sides).



Fig. 10 Temperature History of Case 3 (one layer, Plywood).



Fig. 11 Temperature History of Case 4 (Two Layers, One Air Gap, ?rick-Air-Brick)

REFERENCE

1. Harmathy, T. Z., "A Treatize on Theoretical Fire Endurance Rating," ASTM Special Technical Publication No. 301, American Society for Testing and Materials, Philadelphia, 1961, Page 10

2. Lawson, D. I. and McGuire, J. H. "The Solution of Transient Heat Flow Problems by Analogous Electric Networks," Proceedings, Institution of Mechanical Engineers, Vol 167, 1953, Pages 275-287.

3. Robertson, A. F. and Gross, D., "An Electrical Analog Method of Transient Heat Flow Analysis," Journal of Research of the National Bureau of Standards, Vol. 61, 1958, Pages 105-115.

4. Krokosky, E. M., "An Analog Simulator for Evaluating Materials Subjected to the Standard ASTM Fire Curve," Journal of Materials American Society of Testing Materials, Vol. 2, No. 4, 1967, Pages 801-816.

5. Krokosky, E. M., "Fire Intensity-Resistance Evaluation," Journal of the Structural Division Proceedings of the American Society of Civil Engineers, Nov. 1970, Pages 2501-2519.

6. Ozisik, M. N., "Boundary Value Problems of Heat Conduction," 1968, International Textbook Company, Pennsylvania.

7. Kreith, F., "Principles of Heat Transfer", 1967, International Textbook Company, Pennsylvania

8. Fox, L., "Numerical Solution of Ordinary and Partial Differential Equations," Addison-Wesley, Massachusetts, 1962.

9. Nottage, H.B., "The Thermal Properties of Building Materials Used in Heat Flow Calculations", Amer. Soc. of Heating and Vent. Engineers Research Bulletin, Vol. 53, No. 2, 9, 1947.

FIG. 8 FLOW CHART OF TRANSIENT HEAT TRANSFER THROUGH COMPOSITE WALLS WITH ARBITRARY AIR GAPS



```
C  INPUTS
C   NN= NUMBER OF SOLIDS
C    N=  NUMBER OF NODES IN SOLID
C   M  =  NUMBER OF AIR  SPACES
C  ITEST  = AN INTEGER THAT INDICATES WHETHER FORMULA 6 OR FORMULA 8
C  WILL BE USED IN COMPUTATIONS;   1 INDICATES FORMULA 6 ANY OTHER
C  NUMBER INDICATES FORMULA 8
C   TO  =  INITIAL TEMPERATURE NOW 68
C  GM1 = HEAT RELEASE PER UNIT VOLUME DUE TO MOISTURE VAPORIZATION FOR
C    GYPSUM
C  GM2 = HEAT RELEASE PER UNIT VOLUME DUE TO MOISTURE VAPORIZATION FOR
C    BRICK
C  GM3 = HEAT RELEASE PER UNIT VOLUME DUE TO MOISTURE VAPORIZATION FOR
C    WOOD
C   ID  AN ARRAY OF INTEGERS THAT INDICATES THE SOLID-AIR CONFIGURATION
C THATS BEING TESTED  0= AN AIR GAP  NON-ZERO = SOLID
C   A DATA CARD WITH 1010101  WOULD INDICATE 4 SOLIDS AND THREE AIR GAPS
C  IDD AN ARRAY OF INTEGERS THAT INDICATE THE TYPE OF SOLIDS  1=  GYPSUM
C  2=  BRICK  3=  WOOD  SUPPOSE THERE WERE 4 SOLIDS GYPSUM,WOOD,BRICK
C   AND GYPSUM  THEN DATA CARD WOULD BE PUNCHED AS 1321
C  AL  =  SOLID THICKNESS IN FEET
C  ALPHA =  HEAT DIFUSSION COEFFICIENT
C   RHO  =  DENSITY OF SOLID IN CUBIT FEET
C   C=SPECIFIC HEAT CAPACITY OF SOLID
C   AI = AIR GAP DISTANCE IN FEET
C   GA = HEAT RELEASE PER UNIT VOLUME IN AIR GAP
C  OUTPUTS
C  T(I,J) PRIME  = TEMPERATURE AT TIME(THETA+DTHETA) OF JTH NODE IN
C  ITH SOLID LAYER
C   TA(I)  =  TEMPERATURE OF MID-POINT OF AIR SPACING BETWEEN ITH AND
C  I+1 LAYER OF SOLID   THETA = TIME
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      DOUBLE PRECISION T11,T22,T33,T44,T66,T88 ,T55
      REAL  TO,        AL,ALPHA,RHO,C,AI,GA,   X,Y ,GM1,GM2,GM3
      DIMENSION T(20,10),AL(20),AI(20),TA(20),DELTAX(20),AM(20),H(20),
     1 BH(20),AK(20,10),G(20,10),EPSLON(20   ),RHO(20),C(20),S(20),
     2 P(20),R1(20), Q(20),ID(80) ,GA(20),ALPHA(20)
      DIMENSION X(500,5),Y(500,5),NRMX(5)              , A(20),B(20),
     1 RHOA(10),D(20,10),IDD(20),THETA1(20,10),THETA2(20,10)
      NA=0
C   THETA = TIME IN HOURS
      THETA=0.0D0
C  EP AND EPN = EMISSIVITY
      EP=.835D0
      EPN=.9D0
C  HH=CHARACTERISTIC HEAT TRANSFER COEFFICIENT
      HH=.15D0
```

```
C    CPA= SPECIFIC HEAT CAPACITY FOR AIR
      CPA=0.25D0
C   AMO= VOLUME AIR FLOW THROUGH THE LAYERS, CUBIC FEET PER HOUR
      AMO=  0.D0
C   ROA1= DENSITY OF AIR IN LB PER CUBIC FOOT
      ROA1=.05D0
C   SIGMA = BOLTZMAN STEFAN CONSTANT
      SIGMA=0.1714D-8
C   AK = HEAT CONDUCTION COEFFICIENT
C        READ IN DATA THAT DESCRIBE PANELS AND READ IN OTHER DATA
      READ 90,NN,N,M,ITEST,TO,        GM1,GM2,GM3
      KK=NN+M
      READ 91,(ID(I),I=1,KK)
      READ 91,(IDD(I),I=1,NN)
      READ 92,(AL(I),ALPHA(I)      ,RHO(I),C(I),I=1,NN)
      IF(M.EQ.0)GO TO 307
      READ 97,(AI(I),GA(I),I=1,M)
C   DTHETA = TIME INCREMENT
307   DTHETA=1.D0/ 720.D0
      DO 301 J=1,N
      DO 301  I=1,NN
301   T(I,J)=TO
      IF(M.EQ.0) GO TO 308
      DO 302  I=1,M
302   TA(I)=TO
308   DO 305 I=1,NN
      H(I)=HH
305   EPSLON(I)=EP
      EPSLON(NN)=EPN
      EPSLON(1)=EPN
      DO 104 I=1,NN
104   DELTAX(I)=AL(I)/FLOAT(N-1)
      DO 105 I=1,NN
      AM(I)=DELTAX(I)**2/(ALPHA(I)*DTHETA)
      BH(I)=(2.*H(I)*DTHETA)/(RHO(I)*C(I)*DELTAX(I))
105   R1(I)=(2.*DTHETA*SIGMA*EPSLON(I     ))/(DELTAX(I)*RHO(I)*C(I))
309   PRINT 100,NN,N,M,ITEST,TO,        GM1,GM2,GM3
      PRINT 101,(ID(I),I=1,KK)
      PRINT 101,(IDD(I),I=1,NN)
      IF(M.EQ.0)GO TO 310
      PRINT 102,(AI(I),GA(I),I=1,M)
310   PRINT 102,(AL(I),ALPHA(I)      ,RHO(I),C(I),I=1,NN)
      PRINT 311,(DELTAX(I),AM(I),BH(I),R1(I),H(I),I=1,NN)
      PRINT 102,  (EPSLON(I),I=1,NN)
      G1=GM1
      G2=GM2
      G3=GM3
      TS=TO
      TF=TO
C   PLOT VALUES
      X(1,1)=THETA*60.D0
      X(1,2)=THETA*60.D0
      X(1,3)=THETA*60.D0
      X(1,4)=THETA*60.D0
      X(1,5)=THETA*60.D0
      Y(1,1)=(TF)
      Y(1,2)=(TA(1))
      Y(1,3)=(T(2,N))
```

```
      Y(1,4)=( TA(2))
      Y(1,5)=(T(4,N))
      II=2
      PRINT 87
      PRINT 88
      PRINT 89,TF,T(1,1), TA(1),T(2,N),TA(2),T(3,1), TA(3),T(4,N),THETA
      MM=N-1
1     I=1
      IA=1
      DO 106 JJ=1,N
      DO 106 LL=1,NN
106   G(LL,JJ)=0.0D0
      K=0
C    COMPUTE TF     FORMULA 7
      IF(THETA.GE.1.D0)GO TO 415
      TF=T0+101520.D0*THETA/(60.D0*THETA+4.D0)
      GO TO 417
415   IF(THETA.GE.1.9D0)GO TO 416
      TF=(926.D0+42.D0*THETA-0.0131D0* (120.D0-60.D0*THETA)**2)*1.8D0
     1 +32.D0
      GO TO 417
416   TF=(926.D0+42.D0*THETA)*1.8D0+32.D0
C  COMPUTE T(1,1)  AND TPRIME
417   IJ=IDD(1)
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T(  1 ,1),AK(  1,1),THETA1(1,1),THETA2(1,1),THETA,
     1 DTHETA,G(1,1),G1,G2,G3,IJ)
924   AKK=AK(1,1)
      TPRIME =T(1,1)+R1(1) *(TF-T(1,1))*(TF+T(1,1)+920.D0)*((TF+460.D0)
     1 **2+(T(1,1)+460.D0)**2)+BH(1)*DABS(TF-T(1,1))**.25*(TF-T(1,1))
     2 + (G(1,1)*DELTAX(1)**2)/(AKK*AM(1)) -2.D0/AM(1)*(T(1,1)-T(1,2))
      T(1,1)=TPRIME
C   COMPUTE TPRIME(I,J)  FORMULA 1
5     IF(AM(I).LE.2.D0)GO TO 77
      IJ=IDD(I)
      DO 10 J=2,MM
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T(  I ,J),AK( I,J ),THETA1(I,J),THETA2(I,J),THETA,
     1 DTHETA,G(I,J),G1,G2,G3,IJ)
522   AKK=AK(I,J)
      T(I,J)=T11(AM(I),T(I,J-1),T(I,J+1),T(I,J),G(I,J),AKK    ,
     1 DELTAX(I))
10    CONTINUE
     IF(I.EQ.NN)GO TO 45
      IF(ID(IA+I).EQ.0)GO TO 12
C   SOLID TO SOLID COMPUTATION
      IJ=IDD(I)
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T(  I ,N),AK( I,N ),THETA1(I,N),THETA2(I,N),THETA,
     1 DTHETA,G(I,J),G1,G2,G3,IJ)
932   AKK=AK(I,N)
      DO 875 JK=1,NN
      A(JK)=(RHO(JK)*C(JK)*DELTAX(JK))/2.D0
      B(JK)=(DELTAX(JK)*DTHETA)/2.D0
      D(JK,N)=(AK(JK,N)*DTHETA)/DELTAX(JK)
875   D(JK,1)=(AK(JK,1)*DTHETA)/DELTAX(JK)
      T(I,N)=T55(A(I),T(I,N),A(I+1),T(I+1,1),B(I),G(I,N),B(I+1),
     1 G(I+1,1),D(I,N),T(I,N-1),D(I+1,1),T(I+1,2))
```

```
      T(I+1,1)=T(I,N)
      I=I+1
      GO TO 5
C   COMPUTE TPRIME FORMULA 2  AIR SPACE COMPUTATION
12    IF(AM(I).LE.2.D0)GO TO 75
      K=K+1
      IJ=IDD(I)
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T(  I ,N),AK( I,N ),THETA1(I,N),THETA2(I,N),THETA,
     1 DTHETA,G(I,N),G1,G2,G3,IJ)
611   AKK=AK(I,N)
      T(I,N)=T22(AM(I),T(I,N),T(I,N-1),R1(I),T(I+1,1),BH(I),TA(K),
     1 G(I,N),AKK,DELTAX(I))
C  COMPUTE TAPRIME  FORMULA 3
      RHOA(K)=39.674D0/(TA(K)+460.D0)
      P(K)=(H(K)*DTHETA)/(AI(K)*CPA*RHOA(K))
      GA(K)=0.0D0
      S(K)=(AMO*DTHETA)/(AI(K))
      Q(K)=(H(K+1)*DTHETA)/(AI(K)*CPA*RHOA(K))
      GAA=GA(K)
      TA(K)=T33(TA(K),P(K),T(I,N), Q(K),T(I+1,1),S(K),GAA  )
C  COMPUTE TPRIME(I+1,1)  FORMULA 4
      IF(AM(I+1).LE.2.D0)GO TO 80
      IJ=IDD(I+1)
     T(I+1,1)=T44(T(I+1,1),R1(I),T(I,N),AM(I+1),BH(I+1),TA(K),T(I+1,2),
     1 G(I+1,1),AKK,DELTAX(I+1))
      IF(K.EQ.M)K=0
39    IF(I.EQ.NN)GO TO 45
      I=I+1
      IA=IA+1
      GO TO 5
C     DECIDE BETWEEN FORMULAS 6 AND 8
45    IF(ITEST.NE.1)GO TO 37
C   COMPUTE T(NN,N)  FORMULA 6
      IF(AM(NN).LE.2.D0)GO TO 78
      T1=T0
         IJ=IDD(NN)
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T( NN,N ),AK( NN,N),THETA1( NN,N),THETA2( NN,N),THETA,
     1 DTHETA,G( NN,N),G1,G2,G3,IJ)
431   AKK=AK(NN,N)
      T(NN,N)=T66(T(NN,N),AM(NN),T(NN,N-1),R1(NN),T1,BH(NN),
     1 G(NN,N),AKK,DELTAX(NN))
      GO TO 209
C   COMPUTE T(NN,N)   FORMULA 8
37    IF(AM(NN).LE.2.D0)GO TO 79
      IJ=IDD(NN)
C   TEST PROPERTIES FOR CURRENT SOLID
      CALL PROP(T( NN,N ),AK( NN,N),THETA1( NN,N),THETA2( NN,N),THETA,
     1 DTHETA,G( NN,N),G1,G2,G3,IJ)
432   AKK=AK(NN,N)
      T(NN,N)=T88(T(NN,N)  ,T(NN,N-1),TS,AM(NN),BH(NN),
     1 G(NN,N),AKK,DELTAX(NN),R1(NN))
      PS=0.15D0*DTHETA/(0.5D0*CPA*ROA1)
      TS=TS+PS*DABS(T(NN,N)-TS)**.25*(T(NN,N)-TS)
209   IF(MOD(NA,10).NE.0)GO TO 57
      PRINT 89,TF,T(1,1), TA(1),T(2,N),TA(2),T(3,1), TA(3),T(4,N),THETA
C  PLOT VALUES
```

```
      X(II,1)=THETA*60.D0
      X(II,2)=THETA*60.D0
      X(II,3)=THETA*60.D0
      X(II,4)=THETA*60.D0
      X(II,5)=THETA*60.D0
     Y(II,1)=(TF)
     Y(II,2)=(TA(1))
     Y(II,3)=(T(2,N))
     Y(II,4)=( TA(2))
     Y(II,5)=(T(4,N))
      II=II+1
57    IF(THETA.GT..833D0)  GO TO 200
      NA=NA+1
C   INCREMENT TIME
      THETA=THETA+DTHETA
      GO TO 1
200   II=II-1
      PRINT 87
      NROW=500
      NRMX(1)=II
      NRMX(2)=II
      NRMX(3)=II
      NRMX(4)=II
      NRMX(5)=II
      NARGS=5
      CALL PLOTS(NARGS,X,Y,NRMX,NROW)
      STOP
78    PRINT 81,AM(NN)
      STOP
79    PRINT  82,AM(NN)
      STOP
77     PRINT 83,AM(I)
      STOP
80    PRINT 84, AM(I+1)
      STOP
75    PRINT 85, AM(I)
      STOP
81    FORMAT(1X,'M NOT GREATER THAN 2 FORMULA 6 M=',E14.8)
82    FORMAT(1X,'M NOT GREATER THAN 2 FORMULA 8 M=',E14.8)
83    FORMAT(1X,'M NOT GREATER THAN 2 FORMULA 1 M=',E14.8)
84    FORMAT(1X,'M NOT GREATER THAN 2 FORMULA 4 M=',E14.8)
85    FORMAT(1X,'M NOT GREATER THAN 2 FORMULA 2 M=',E14.8)
88    FORMAT('       TF          T(1,1)        TA(1)         T(2,N)
     1 TA(2)        T(3,1)         TA(3)        T(4,N)      THETA')
101   FORMAT(20I3)
100   FORMAT(1X,4I6,5F11.3)
98    FORMAT(10F6.0)
92    FORMAT( 4F6.0)
97    FORMAT(2F6.0)
91    FORMAT(20I1)
216   FORMAT(1X,1X,2I3,3F14.2)
90    FORMAT(4I2,5F6.0)
89    FORMAT(1X,F12.3,1X,F12.3,1X,F12.3,1X,F12.3,1X,F12.3,1X,F12.3,
     1 F12.3,1X,F12.3,1X,F6.3)
87    FORMAT(1H1)
102   FORMAT(1X,4F14.3)
312   FORMAT(1X,4D26.16)
311   FORMAT(1X,F14.8,4D26.16)
```

```
86      FORMAT(1X,F12.3,1X,F12.3,1X,F12.3,1X,F12.3,13X,F12.3)
317     FORMAT(1X,3F12.3,D26.16,3F12.3)
        END


        DOUBLE PRECISION FUNCTION T11(AM,T1,T2,T3,G,AK,DELTAX)
        IMPLICIT DOUBLE PRECISION(A-H,O-W)
C     FORMULA 1
        T11=(T1+T2)/AM+ (1.D0-2.D0/AM)*T3  +(G*DELTAX**2)/(AK*AM)
2       FORMAT('  FORMULA 1')
        RETURN
        END


C  FORMULA 2
      DOUBLE PRECISION FUNCTION T22(AM,T1,T2,R1,T3,H,TA,G,AK,DELTAX)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      T22=T1+2.D0/AM*(T2-T1)-R1*(T1-T3)*(T1+T3+920.D0)*((T1+460.D0)**2+
     1(T3+460.D0)**2)-H*DABS(T1-TA)**.25*(T1-TA)+(G*DELTAX**2)/(AK*AM)
1     FORMAT(1X,3F12.3,D26.16,3F12.3)
2     FORMAT('  FORMULA 2')
      RETURN
      END


C  FORMULA 3
      DOUBLE PRECISION FUNCTION T33(TA,P,T1,Q,T2,S,GA)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      T33=TA +P*DABS(T1-TA)**.25*(T1-TA)-Q*DABS(TA-T2)**.25*(TA-T2)+
     1 S*(T1-T2)+GA
2     FORMAT('  FORMULA 3')
      RETURN
      END


C  FORMULA 4
      DOUBLE PRECISION FUNCTION T44(T1,R1,T2,AM,H,TA,T3,G,AK,DELTAX)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      T44=T1 +R1*(T2-T1)*(T2+T1+920.D0)*((T2+460.D0)**2+(T1+460.D0)**2)
     1 -2.D0/AM*(T1-T3)+H*DABS(TA-T1)**.25*(TA-T1)+(G*DELTAX**2)/(AK*AM)
2     FORMAT('  FORMULA 4')
      RETURN
      END


C  FORMULA 6
      DOUBLE PRECISION FUNCTIONT66(T1,AM,T2,R1,T0,H,G,AK,DELTAX)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      T66=T1+2.D0/AM*(T2-T1)-R1*(T1-T0)*(T1+T0+920.D0)*((T1+460.D0)**2
     1 +(T0+460.D0)**2)-H*DABS(T1-T0)**.25*(T1-T0)+(G*DELTAX**2)/(AK*AM)
2     FORMAT('  FORMULA 6')
      RETURN
      END
```

```
C  FORMULA 8
      DOUBLE PRECISION FUNCTION T88(T1,T2,TS,AM,H,G,AK,DELTAX,R1)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
      T88=T1+2.D0/AM*(T2-T1)-H*DABS(T1-TS)**.25 *(T1-TS)+(G*DELTAX**2)/
     1(AK*AM)
 2    FORMAT('  FORMULA 8')
      RETURN
      END


C     FORMULA
      DOUBLE PRECISION FUNCTION T55(A1,T1,A2,T2,B1,G1,B2,G2,D1,T3,D2,
     1 T4)
      IMPLICIT DOUBLE PRECISION (A-H,O-W)
      T55=(A1*T1+A2*T2+B1*G1+B2*G2+D1*(T3-T1)-D2*(T2-T4))/(A1+A2)
      RETURN
      END
      SUBROUTINE PROP(T1,AK,THETA1,THETA2,THETA,DDTHET,G,GM1,GM2,GM3,IJ
     1)
      IMPLICIT DOUBLE PRECISION(A-H,O-W)
C  TEST PROPERTIES FOR GYPSUM
      GO TO (501,502,503),IJ
501   IF(    T1.GE.0.0D0.AND.    T1.LE.200.D0)AK=.5D0
      IF(    T1.GE.200.D0.AND.    T1.LE.400.D0)AK=.3D0
      IF(    T1.GE.400.D0.AND.    T1.LE.2000.D0)AK     =.2D0+T1/4000.D0
      AK=AK/4.0
     IF(    T1.GE.212.D0.AND.T1.LE.220.D0    )THETA1=THETA
     IF(    T1.GE.212.D0)THETA2     =THETA-THETA1
      IF(THETA2.LT.DDTHET)GO TO 522
      IF(THETA2     .GE.0.0D0.AND.THETA2    .LE.200.D0*DDTHET)G     =-
     1 (GM1+14100.D0)/(200.D0*DDTHET)
      GO TO 522
C  TEST PROPERTIES BRICK
502   IF(    T1.GE.0.0D0.AND.    T1.LE.200.D0)AK=1.D0
      IF(    T1.GE.200.D0.AND.    T1.LE.2000.D0)AK     =.46D0+2.D-4*T1
     IF(    T1.GE.212.D0.AND.T1.LE.220.D0    )THETA1=THETA
     IF(    T1.GE.212.D0)THETA2     =THETA-THETA1
      IF(THETA2     .LT.DDTHET)GO TO 522
      IF(THETA2     .GE.0.0D0.AND.THETA2   .LE. 100.D0*DDTHET)G     =-
     1  GM2/(100.D0*DDTHET)
      GO TO 522
C     TEST PROPERTIES FOR WOOD,DOUGLAS FIR
503   AK=.065D0
     IF(    T1.GE.212.D0.AND.T1.LE.220.D0    )THETA1=THETA
     IF(    T1.GE.212.D0)THETA2     =THETA-THETA1
      IF(THETA2     .LT.DDTHET)GO TO 522

      IF(THETA2     .GE.0.0D0.AND.THETA2   .LE. 200.D0*DDTHET)G     =-
     1  GM3/(200.D0*DDTHET)
522   RETURN
      END
```

Acknowledgement

The author wishes to acknowledge the contributibution of Mr. Vernon Dantzler, Applications Section, Computer Services Division, for coding the Fortran program, and for his indispensible and cheerful help in modifying and updating the Fortran program.

The author also wishes to thank Mr. J. Newton Breese for his part in preparing the graphs and the flow chart for reproduction.